Perimetric complexity is a measure of the complexity of binary pictures. It is defined as the sum of inside and outside perimeters of the foreground, squared, divided by the foreground area, divided by 4\pi . Difficulties arise when this definition is applied to digital images composed of binary pixels. In this paper we identify these problems and propose solutions. Perimetric complexity is often used as a measure of visual complexity, in which case it should take into account the limited resolution of the visual system. We propose a measure of visual perimetric complexity that meets this requirement.

Background

Perimetric complexity is a measure of the complexity of binary pictures. It is defined as the sum of inside and outside perimeters of the foreground, squared, divided by the foreground area, divided by 4\pi. The concept of perimetric complexity was first introduced (and called dispersion) by Attneave and Arnoult [1] in an effort to explain the apparent preceptual complexity of visual shapes. In the field of image processing, the concept appears as its inverse compactness [2-4]. The concept was given new life (and a new name) in 2006 by Pelli et al., who showed that the efficiency of letter identification was nearly proportional to perimetric complexity [5]. It has since become a popular metric in a variety of shape analysis applications, including human letter identification [5-7], handwriting recognition [8], evolution of graphical symbols [9], and design of graphical anti-spam technologies [10-12].

In this note we develop Mathematica functions to compute perimetric complexity of binary digital images, and illustrate their application. The code is compatible with Version 8 of Mathematica.

Although the concept of perimetric complexity is clear when applied to continuous plane shapes, complications arise when the concept is extended to binary digital images. We discuss these complications and suggest suitable solutions. We also introduce the concept of visual perimetric complexity, which takes into account the blurring action of the human visual system.

We begin by illustrating the application of the function PerimetricComplexity to a binary
We begin by illustrating the application of the function \( \text{PerimetricComplexity} \) to a binary image.

\[
\text{cat} = \begin{array}{c}
\end{array}
\]

\[
\text{PerimetricComplexity}[\text{cat}]
\]

\[
\{2110.08, 89588, 3.95494\}
\]

The output is a list containing the perimeter (in pixels), the area (in square pixels), and the complexity. In the following sections we will describe the derivation of this function, as well as the options that may be used to control its operation.

### Perimetric complexity of geometric shapes

Perimetric complexity is a measure of the complexity of binary pictures. In a binary picture, one or several regions of the same color (white) are defined as foreground, and the remainder (black) as background. Perimetric complexity \( C \) is defined here as the sum of inside and outside perimeters of the foreground \( P \), squared, divided by the foreground area \( A \), divided by \( 4\pi \)

\[
C = \frac{p^2}{A} \frac{4\pi}{4\pi}.
\]

In the remainder of this note, unless otherwise noted, we will use the term complexity as synonymous with perimetric complexity.

We begin with the example of a circular disk with unit radius.

\[
\text{Graphics}[[\text{White, Disk[]}, \text{PlotRange} \rightarrow 2, \text{Background} \rightarrow \text{Black}]]
\]

Here the perimeter is \( 2\pi \), and the area is \( \pi \), so the complexity is

\[
C = \frac{(2\pi)^2}{\pi 4\pi} = 1
\]

It can be shown that the disk is the shape with the lowest complexity. The normalizing constant \( 4\pi \) in the definition leads to a unit value for this most simple shape. As a consequence, any other
value of complexity is easily compared to that of the circular disk. Pelli et al. [5] suggest that complexity is closely related to the number of visual features in a shape. In that sense, we could say that the circular disk has only a single feature.

Our next example is a square with unit sides.

\[
\text{Graphics[}{\{\text{White, Rectangle[-\{1, 1\}, 2, \{1, 1\}, 2]}}\text{, PlotRange }\rightarrow 1, \\
\quad \text{Background }\rightarrow \text{Black}\text{]}\]

The perimeter here is 4, and the area is 1, so the perimetric complexity is.

\[
C = \frac{4^2}{4 \pi} = \frac{4}{\pi} \approx 1.27
\]

If we add a square hole in the center, with sides 1/2, there will be an interior perimeter as well, as shown here.

\[
\text{Graphics[}{\{\text{White, Rectangle[-\{1, 1\}, 2, \{1, 1\}, 2]}}\text{, Black, Rectangle[-\{1, 1\}, 4, \{1, 1\}, 4]}}\text{, PlotRange }\rightarrow 1, \\
\quad \text{Background }\rightarrow \text{Black}\text{]}\]

Now the total perimeter is the sum of inner and outer perimeters, and the area is the difference in areas of the squares, so

\[
C = \frac{(4 + 2)^2}{(1 - 1/4) 4 \pi} = \frac{12}{\pi} \approx 3.82
\]

So according to this measure, the square with a hole is about three times as complex as the square.

Some important observations about complexity are 1) it is dimensionless, 2) it is independent of scale or orientation, 3) it is additive. By additive we mean that the complexity of a pair of shapes, considered as a single shape, is equal to the sum of their PCs computed separately.
**Perimetric complexity of plane curves**

Although it is beyond the scope of this note, we note for reference that if a shape is defined by a closed parametric curve, its exact complexity can be obtained using calculus methods [13]. Specifically, if over an interval \(a \leq t \leq b\) the functions \(x(t)\) and \(y(t)\) and their derivatives \(\dot{x}(t)\) and \(\dot{y}(t)\) are continuous, then the curve described has a length

\[
P = \int_{a}^{b} \sqrt{\dot{x}^2 + \dot{y}^2} \, dt
\]

and an area

\[
A = -\int_{a}^{b} \dot{y} \, x \, dt.
\]

**Perimetric complexity of binary digital images**

A digital image is defined here as a rectangular array of square pixels. A binary digital image contains pixel values of 1 (white) and 0 (black) only. The foreground consists of the white pixels. The original definition of complexity relies upon the notion of a perimeter, which has no unique analog in the context of digital images. However, two definitions of perimeter are available, as described below.

- Using “PerimeterLength”

The first definition we consider is the most straightforward. Consider a binary image consisting of a single white pixel:

\[
\text{ImagePad[Image[{{1}}], 1]}
\]

It seems natural to define the perimeter of this shape as 4, and the area as 1, so \(C = 4/\pi\), the same as the square discussed earlier.

Now consider this shape consisting of 3 white pixels.

\[
\text{ImagePad[Image[{{1, 0}, {1, 1}}], 1]}
\]

Here the perimeter, consisting of the exposed pixel faces, is 8, and the area is 3, so \(C = 16/(3 \pi)\).

Extending this idea, we can define the perimeter as the sum of the exposed faces of pixels in the foreground.

Version 8 of *Mathematica* includes a set of operators from the discipline of mathematical morphol-
ogy. These can be used to easily calculate perimetric complexity. To illustrate this we begin with a binary image with several separated parts.

\[ y_i = \text{以} \]

The `MorphologicalComponents` operator finds connected regions, and labels them with integers. The `Colorize` operator visualizes these regions by assigning colors to each label. The `CornerNeighbors -> False` option ensures that only 4-connected neighborhoods are considered.

```
MorphologicalComponents[yi, CornerNeighbors -> False] // Colorize
```

The `ComponentMeasurements` operator returns a selected set of measurements about each region. In this case we are interested in the area and the perimeter length. The results are returned as a set of rules, showing the results for each region.

```
cm = ComponentMeasurements[
    MorphologicalComponents[yi, CornerNeighbors -> False],
    {"PerimeterLength", "Count"}, CornerNeighbors -> False]

{1 -> {56, 98}, 2 -> {198, 469}, 3 -> {138, 371}}
```

We can combine the perimeters and areas of the several regions, and then compute complexity in the usual way.

```
Total[Last /@ cm]

{392, 938}

% [[1]]^2 /%[[2]] / (4 Pi) // N

13.0365
```

The preceding calculations are implemented in the function `PerimetricComplexity`, defined in the Appendix. To obtain the previous result, we evaluate:

```
PerimetricComplexity[yi, Filter -> None,
    Method -> "PerimeterLength"] // N

{392., 938., 13.0365}
```

The option `Method -> "PerimeterLength"` ensures that we use the definition of perimeter described above. The option `Filter -> None` will be explained below.
For future reference, to distinguish it from variants that we will consider, we will call this the “raw” perimetric complexity.

- Using “PolygonalLength”

The second definition of the perimeter of a connected region in a binary digital image is to consider the perimeter pixel centers as points on a lattice, and to define the length as the sum of the sides of the polygon defined by those points. This estimate of the perimeter is obtained from ComponentMeasurements by using the measurement “PolygonalLength”.

```mathematica
cm = ComponentMeasurements[
    MorphologicalComponents[yi, CornerNeighbors -> False],
    {"PolygonalLength", "Count"}, CornerNeighbors -> False
] // Total // N
```

Note that the measures of perimeter length are smaller than before.

And again complexity can be easily computed:

```mathematica
N[#[[1]]^2 / #[[2]] / (4 Pi)] &@Total[Last /@ cm]
```

8.47953

This variant of perimetric complexity is implemented with the following options:

```mathematica
PerimetricComplexity[yi, Filter -> None,
    Method -> "PolygonalLength"] // N
```

{316.149, 938., 8.47953}

Or, since Method->"PolygonalLength" is the default,

```mathematica
PerimetricComplexity[yi, Filter -> None] // N
```

{316.149, 938., 8.47953}

- Approximating complexity of continuous shapes

We introduced the concept of perimetric complexity with a few continuous shapes, such as a square and a circle. In these cases, complexity is easily calculated, because we have simple formulas for the area and perimeter. It might be imagined that complexity of the continuous shape could be approximated by computing the complexity of a discrete sampled image, rendered from the shape. As we shall see, this assumption is not strictly correct.

Consider the circular disk. As noted at the beginning, it has \( C = 1 \).
Now we consider an image rendered from the continuous shape. We render it into a certain size image. We change the color so that the foreground is white, as is our convention.

```
disk = Graphics[{White, Disk[]}, Background -> Black]
```

If we compute the complexity, we find that it is too large, by 62%, relative to the continuous shape.

```
size = 2^7 + 1;
diskimage = Image[disk, “Bit”, ImageSize -> {1, 1} size,
                   ColorSpace -> "Grayscale"]
```

The reason is that the sampled image actually is more complex than the continuous shape. Its contour is jagged, while that of the continuous shape is smooth. It might be imagined that this could be remedied by increasing the resolution of the rendering. Here we show that belief is misplaced. We render at several sizes, and plot the results. Size has little effect, and the complexity never approaches the value of 1 corresponding to the continuous shape.
This problem can be somewhat ameliorated by using the \textbf{PolygonalLength} measure of perimeter length. Rather than the pure approach of measuring the exposed face of each foreground pixel, this measures the length of the contour that travels between the centers of those pixels.

\begin{verbatim}
PerimetricComplexity[diskimage, Filter -> None]
\end{verbatim}

\{403.647, 11873, 1.09203\}

Now the difference is reduced to 9%. Here again, the reader might think that this difference could be reduced to zero by enlarging the resolution (number of pixels) in the rendered image, but this is not so. We leave that as an exercise for the reader. The error can never be zero, because the path between pixel centers must always be vertical, horizontal, or diagonal, so it can never smoothly follow the true circular contour. Put another way, it has a higher fractal dimension than the circle, and thus greater length.

\section*{Pelli algorithm}

Pelli \textit{et al.} \cite{Pelli} proposed a method for computing complexity which we quote here in full:

The ink area is the number of 1’s. To measure the perimeter we first replace the image by its outline. (We OR the image with translations of the original, shifted by one pixel left; left and up; up; up and right; right; right and down; down; and down and left; and then bit clear with the original image. This leaves a one-pixel-thick outline.) It might
seem enough to just count the 1’s in this outline image, but the resulting “lengths” are non-Euclidean: diagonal lines have “lengths” equal to that of their base plus height. Instead we first thicken the outline. (We OR the outline image with translations of the original outline, shifted by one pixel left; up; right; and down.) This leaves a three-pixel-thick outline. We then count the number of 1’s and divide by 3.

This method can be implemented using the Dilation operator, as we show here. With verbose == True, it shows two images: the perimeter in red, and the thickened perimeter. It returns the length of the perimeter, the area, and the complexity.

```
2/28/11 12:10:52 (Local) In[2]:=
PelliMethod[image_, verbose_: False] := Module[
  {tmp0, tmp1, tmp2, tmp3, perimeter, area},
  tmp0 = ImagePad[image, 2];
  tmp1 = Dilation[tmp0, BoxMatrix[1]];
  tmp2 = ImageSubtract[tmp1, tmp0];
  tmp3 = Dilation[tmp2, CrossMatrix[1]];  
  If[verbose, (Print@ GraphicsRow[
    {ColorCombine[{{tmp1, tmp0, tmp0}}, tmp3]}]);]
  perimeter = Total[ImageData[tmp3], 2]/3;
  area = Total[ImageData[tmp0], 2];
  {perimeter, area, perimeter^2/area/(4 Pi)}
]
```

Applying this to the three-component Chinese character, we get:

```
PelliMethod[yi, True] // N
```

```
(326.333, 938., 9.03463)
```

The exact method gives the following:

```
PerimetricComplexity[yi, Filter -> None,
  Method -> "PerimeterLength"] // N
```

```
(392., 938., 13.0365)
```

We see that the perimeter is substantially underestimated by the Pelli method in this case. This method has other limitations. It effectively assumes regions that are large in pixel dimensions. For example, consider the case of a single pixel object. As noted above, it has C = 4/π ≈ 1.273. But the Pelli method yields an complexity value more than three times too large.
Visual perimetric complexity

Much of the motivation for the use of perimetric complexity is the hope that it might provide an approximate measure of the visually perceived complexity of shapes. But this only makes sense if the shape is actually visible. Consider the difference between the continuous circular disk, and its sampled image, as discussed above. They have different perimetric complexities, no matter how high the resolution of the sampled version. But of course, at a certain viewing distance, they will be indistinguishable.

Filtering

Here we propose an approach to dealing with this problem. The idea is to first blur the image, in a manner consistent with visual blur, and then compute perimetric complexity. We begin with the example of a Chinese character.

To make things simple, we will use Gaussian blur, although this is not an accurate description of human visual blur. Later we will show a more accurate form of blur.

First we pad the image, so that the blur is contained.
tmp0 = ImagePad[jun, 5]

Next we magnify the image. This allows greater flexibility in the filtering.

mag = 4;
tmp1 = ImageResize[tmp0, mag ImageDimensions[tmp0]]

Then we blur the image, in this case by a Gaussian filter with a radius of 8 pixels.
tmp2 = GaussianFilter[tmp1, 8]

Then we binarize the image. Unfortunately, this requires some method of setting the threshold. Here we use a fixed threshold of 0.5.

tmp3 = Binarize[tmp2, 0.5]

And because we imagine that the image is viewed at such a distance that the pixels are not resolved, we use the PolygonalLength method.
PerimetricComplexity[tmp3, Filter -> None, Method -> "PolygonalLength"] // N

\{1944.82, 24044., 12.5182\}

We can compare this to the unfiltered “raw” version:

PerimetricComplexity[jun, Filter -> None, Method -> "PerimeterLength"] // N

\{606., 1467., 19.9207\}

The filtered version has substantially lower complexity, as we expect.

- Visual filtering using a Gaussian

For the filtering to approximate visual blur, it must be based on the size of the original image and its distance from the viewer. Obviously, as the shape becomes smaller or farther from the observer, its details are more blurred, less visible, and contribute less to the visual complexity.

The challenge is to determine the appropriate value of the Gaussian filter radius for a given viewing distance. From measurements of visual sensitivity, we know that visual Gaussian blur has a standard deviation of about $B = 0.01549$ degrees of visual angle [11]. But we need to convert this into a radius in pixels. Recall that the image may be magnified by $M$ before filtering. If we express the viewing distance $V$ in terms of pixels (before magnification), then the size of each pixel $S$ (after magnification) is approximately

\[
S = \frac{57.3}{M V} \text{ deg}
\]

The constant 57.3 is an approximation to $1/\text{ArcTan}[1^\circ]$.

\[
1 / \text{ArcTan}[1 \text{ }^\circ]
\]

57.3016

By default, the radius is twice the standard deviation. So the filter radius should be

\[
R = \frac{2B}{S} = \frac{2BMV}{57.3} \text{ pixel}
\]

Consider an example. Suppose that charimage is displayed on a typical computer screen with a resolution of 72 pixels/inch, and viewed at a distance of 12 inches. Then

\[
V = 12 \times 72
\]

864

And recall that

\[
B = 0.01549;
\]
Suppose further that the magnification is

\[ M = 4; \]

Then

\[ R = \frac{2BMV}{57.3} = 1.86853 \]

proceeding now with the steps outlined above for filtering,

\[
\begin{align*}
tmp0 &= \text{ImagePad}[\text{jun}, 1]; \\
tmp1 &= \text{ImageResize}[tmp0, M \text{ImageDimensions}[tmp0]]; \\
tmp2 &= \text{GaussianFilter}[tmp1, R]; \\
tmp3 &= \text{Binarize}[tmp2, 0.5] \\
\text{PerimetricComplexity}[tmp3, \text{Filter} \rightarrow \text{None}] \end{align*}
\]

\[
[2365.71, 23472., 18.9741]
\]

Notice that the final result is not that different from the raw value.

\[
\begin{align*}
\text{PerimetricComplexity}[\text{jun}, \text{Filter} \rightarrow \text{None}, \\
\quad \text{Method} \rightarrow \text{"PerimeterLength"}] \rightarrow N \\
\end{align*}
\]

\[
[606., 1467., 19.9207]
\]

This is because the image is viewed close enough that the blur has little effect. But if we repeat the process, with 4 times the viewing distance, we will see that the complexity is substantially reduced.

\[
\frac{1}{\text{ArcTan}[1. ^\circ]} = 57.3016
\]
\[ \frac{2.33}{60} \]

\[ 0.0388333 \]

\[ V = 4 \times 12 \times 72; \]
\[ M = 4; \]
\[ R = \frac{2BMV}{57.3} \]
\[ \text{tmp0} = \text{ImagePad}[\text{jun}, 1]; \]
\[ \text{tmp1} = \text{ImageResize}[\text{tmp0}, \text{M ImageDimensions}[\text{tmp0}]]; \]
\[ \text{tmp2} = \text{GaussianFilter}[\text{tmp1}, \text{R}]; \]
\[ \text{tmp3} = \text{Binarize}[\text{tmp2}, 0.5] \]
\[ \text{PerimetricComplexity}[\text{tmp3}, \text{Filter} \rightarrow \text{None}] // \text{N} \]

\[ 7.47413 \]

\[ \{1987.85, 23932. , 13.1395\} \]

The preceding steps of padding, magnification and filtering are built in to the function \texttt{PerimetricComplexity}, as shown below. With \texttt{Verbose->True}, the function also shows the original, the filtered version, the binarized version, and the original (in red) with the perimeter of the filtered version (in white - within the original- and aqua - outside the original).
PerimetricComplexity[jun, Magnification -> 4, Filter -> "Gaussian", ViewingDistance -> 4 x 12 x 72, Verbose -> True] // N

\[
\text{\{1987\,85, 23\,932., 13\,1395\}\n\]

The Gaussian is parameterized by a scale in degrees of visual angle. The default value is 2.33/60 degrees. The user can experiment with different values via the option \text{GaussianScale}.

\section*{Accurate visual filtering with Sech}

Visual blur is more accurately represented with filters other than a Gaussian. In one simple form, the kernel is a Sech (Hyperbolic Secant) function \cite{14,15}. This filter can be selected with an option (the default) in the function \text{PerimetricComplexity}:

\[
\text{PerimetricComplexity}[\text{jun, Filter \to \text{"Sech"}]} // N
\]

\[
\text{\{1016.14, 5977., 13.7472\}\n\]

Or, because it is the default,

\[
\text{PerimetricComplexity}[\text{jun}] // N
\]

\[
\text{\{1016.14, 5977., 13.7472\}\n\]

The Sech is parameterized by a scale in degrees of visual angle. The default value is 2.16/60 degrees \cite{14,15}. The user can experiment with different values via the option \text{SechScale}.

To the advanced reader, it may be of interest that the Sech kernel corresponds to a Sech Contrast Sensitivity Function, with a scale that is the inverse of the kernel. For example the CSF corresponding to the default scale is plotted here:
Accurate visual filtering with an arbitrary point spread function

In real human eyes, blur results from both low-order aberrations such as defocus and astigmatism, and also from higher-order aberrations. In this example, we use a blur function defined by an array of values representing the filter kernel. This example is an actual estimate of the point-spread function for an individual human observer, as measured using a device called a wavefront aberrometer, that includes both low and high order aberrations [16].

```
ImageAdjust[Image[psf]]
```

We can use this blur kernel by supplying it directly to the Filter option.
We will not address this topic in detail, but when a kernel is supplied, the pixels of the kernel and of the magnified image must be of the same size, in degrees of visual angle, for the filter to be accurate. The size of the magnified pixels in degrees will depend upon the viewing distance and the magnification, as described above. Here we magnify the image, and get a different result.

\[
\text{PerimetricComplexity[charimage, Verbose \rightarrow True, Filter \rightarrow psf,}
\]
\[
\text{Magnification \rightarrow 4] // N}
\]

\[
(1405.26, 25663., 6.12344)
\]

**Magnification parameter**

The Magnification parameter should be set with a value that ensures that a filter kernel, if present, has enough pixels in it to adequately represent the filter. For the Sech filter, we generally want a width of at least 3 times the scale, and at least 8 pixels. This means that the magnification should be greater than

\[
M \geq \frac{8 \times 57.3}{3 V B}
\]

(9)

This rule is effectively implemented by the default Magnification\(-\rightarrow\)Automatic option.

**Binarization**

After applying visual blur, it is necessary to binarize the image before calculation of perimetric complexity. There are many ways to binarize an image, and mathematica offers many of them as
options. The simplest is to use a fixed threshold. Since our images are initially defined as 0 or 1, a natural choice of threshold is 0.5. One drawback of this choice is that as images become severely blurred, no pixels may remain that exceed the threshold. From a perceptual point of view, the mean might appear a reasonable choice. As the image blurs, all pixels revert towards the mean, but some always remain above the mean until a uniform image is reached. A drawback of the mean is that it is influenced by the area of the background. For this reason, we adopt the fixed value of 0.5 as the default threshold for binarization.

We should acknowledge that a more valid visual thresholding scheme might be devised, that better reflects our perceptual segregation of areas into light and dark. This is a topic for future research.

It should also be acknowledged that for severely blurred images, considerable grayscale information remains that is lost in binarization. Thus we should question whether perimetric complexity is an appropriate measure for such images.

To illustrate the problem, we show an example of a character viewed at 10 feet on a display with 100 pixels/inch. Note that the blurred image displays internal grayscale structure that is not conveyed by the binarized version.

```mathematica
PerimetricComplexity[charimage, ViewingDistance -> 10 × 12 × 100,
                      Verbose -> True] // N
```

![Image of a Chinese character viewed at different distances](image)

\[199.51, 1604., 1.97476]\]

- **Viewing distance**

For a given shape, complexity will decline with viewing distance, as a result of visual blur. Here we illustrate this effect with an example Chinese character. First we find the raw complexity,

```mathematica
raw = PerimetricComplexity[jun, Filter -> None,
                           Method -> "PerimeterLength"][[3]] // N
19.9207
```

Now we compute complexity for viewing distances ranging from 3 inches to 10 feet, assuming a display with a resolution of 100 pixels/inch. For reference, we show as red lines the raw complexity, and the theoretical limit of 1 (a circular disk). As it should, the visual complexity proceeds from one of these limits to the other as distance increases.
At very small viewing distances (in pixels) the blur has little effect on each pixel, so the visual complexity approaches the raw value. As a rule of thumb, this asymptote is approached when the size of each pixel exceeds 1/4 degree.

It is reassuring to know that the algorithm does approach the correct asymptote as distance increases. Here we show the intermediate images for a case of extreme blur (distance = 30 feet).

\begin{verbatim}
PerimetricComplexity[charimage, Verbose \[ \rightarrow \] True, ViewingDistance \[ \rightarrow \] 30 \times 12 \times 100]
\end{verbatim}

\begin{verbatim}
{104.669, 818, 1.06579}
\end{verbatim}

But a note of caution is warranted. Consider the effect of distance on our other example character
yi (اختبار). We show it here along with the previous plot. Note that the curves cross, so that at large distances (large blurs) the “simpler” character becomes the more complex of the two.

```mathematica
points2 = 
   (#, PerimetricComplexity[yi, ViewingDistance -> # 100][[3]]) & /@ 
   Range[min, max, 3] // N;
ListPlot[{points, points2}, Frame -> True, Joined -> True, 
   Epilog -> {Red, Line[{{min, raw}, {max, raw}}], 
     Line[{{min, 1}, {max, 1}}]}, 
   PlotRange -> {{0, Automatic}, {0, 22}}
   , PlotStyle -> {Blue, Green}, Axes -> False, 
   FrameLabel -> {"Distance (inches)"", "complexity"}]
```

This makes sense, since the densely packed features of the “complex” character blur onto each other, while the more widely separated features of the “simple” character remain distinct. This is illustrated in the following, which shows the intermediate images for the two characters when highly blurred (distance = 8 feet).
But the conclusion we must draw is that even the relative complexity of different shapes cannot be known without specifying the viewing distance.

As a default value, we use the quantity \( \text{ViewingDistance} \rightarrow 48 \div \text{ArcTan}[1^\circ] \). This corresponds to a visual resolution of 48 pixels/degree. It is commonly encountered resolution, about that achieved by a display with 100 pixels/inch viewed from 27 inches. But in actual use, it is advised to use the actual viewing distance rather than relying on this default.

Here is a panel that allows you to experiment with different viewing distances. The complexity, and the diagnostic images are shown.

\[
\text{Manipulate}[
\{\#[[3]] \div N, \text{GraphicsRow}[\#[[4]], \text{ImageSize} \rightarrow 400]\} & @
\text{PerimetricComplexity}[\text{charimage}, \text{ReturnImages} \rightarrow \text{True},
\text{ViewingDistance} \rightarrow \text{inches} 100]
\}, \{\text{inches}, 1, 200\}]
\]

**Recommended practice**

The reader, and potential user of \textbf{PerimetricComplexity} may be daunted by the number of variants of the operator, or unsure what parameters to use. Here we offer suggestions that simplify this matter.

In general, we recommend using the function with default parameters. As a reminder, these are:
Options[PerimetricComplexity]

{Magnification \to \text{Automatic}, \text{Pad} \to \text{Automatic},
\text{Verbose} \to \text{False}, \text{Threshold} \to 0.5, \text{Method} \to \text{PolygonalLength},
\text{Filter} \to \text{Sech}, \text{Normalized} \to \text{True}, \text{ViewingDistance} \to 2750.48,
\text{MorphologicalOperators} \to \text{True}, \text{SechScale} \to 0.036,
\text{GaussianScale} \to 0.0388333, \text{ReturnImages} \to \text{False}}

The only option that should be specified for the typical use of this function is ViewingDistance. This specifies the distance from the eye to the image, specified in units of pixels. The default is ViewingDistance\to 2750.48, consistent with a display having 96 pixels/inch, viewed at 28.65 inches (equal to a display with 48 pixels/degree of visual angle).

It is difficult to imagine a case in which vision, in some form, would not be used to view the shape in question. If such a case arises, however, the “raw” complexity can be measured with the following options:

\text{SetOptions}[\text{PerimetricComplexity, Filter} \to \text{None},
\text{Method} \to \text{"PerimeterLength"}]

This will also be an appropriate measure when the pixels are very large (larger than 1/4 degree).

When trying to approximate the raw complexity of a continuous shape by means of a sampled representation (e.g., a circle via an image of a circle), the following options will yield the lowest error. But as noted above, the error will still be significant.

\text{SetOptions}[\text{PerimetricComplexity, Filter} \to \text{None}]

\section{Examples}

We conclude with two examples of the application of \text{PerimetricComplexity}. The first example is a set of three binary images. Below each image we print the complexity.

\begin{verbatim}
pictures = \{\text{cat, horse, family}\};
\end{verbatim}
The second example is an array of characters. This array was created as part of an experiment on 
the effect of complexity on visual acuity [6]. The first row are the Sloan letters, a well known set 
of letter acuity targets [5]. The remaining six rows are sets of Chinese characters, selected so as to 
be of equal complexity within a row, but increasing in complexity from row to row [6]. The metric 
of complexity used for selection was different from that developed in this paper.

We first apply the function to each character, and plot the results, with a different curve for each 
set.
The figure shows that there is considerable variation in each set. If we take the mean of each set we see a progression in complexity, except for the last three sets. In this figure we show at the bottom an exemplar from each set.

We have illustrated several different methods for computing the perimetric complexity of binary digital images. These methods differ in how they compute the perimeter, and in whether the image

**Conclusions**

We have illustrated several different methods for computing the perimetric complexity of binary digital images. These methods differ in how they compute the perimeter, and in whether the image
is blurred and binarized before the complexity calculation. We have introduced the concept of visual perimetric complexity, and argued that in general it requires blur for a sensible estimate. We have described several methods of implementing visual blur.

The computed value of visual perimetric complexity depends somewhat upon details of the calculation, such as the presumed magnitude and nature of visual blur, and the binarization threshold. In this regard, we have proposed a set of standard default settings and procedures for calculation of visual perimetric complexity.

We have also made the observation that visual perimetric complexity cannot be estimated without specifying the resolution of the display and the viewing distance. As a general rule, the visual perimetric complexity will approach the raw complexity when the width of a pixel exceeds 1/4 degree of visual angle.

## Appendix

### Functions

Here we define several functions based on the derivations presented above.

- **PerimetricComplexity**

  `Clear[PerimetricComplexity]`
PerimetricComplexity::usage =  
"PerimetricComplexity[image_, opts___Rule] Compute the perimetric complexity of a binary image. Perimetric complexity is defined as the square of the sum of inner and outer perimeters divided by the foreground area, divided by 4\pi. By default, foreground pixels are white (1). The function returns a list: \{perimeter, area, complexity\}. Optionally, the image can be filtered and thresholded before calculation of complexity, to provide a a better estimate of the visual complexity of the image. Options available are Pad (width of padding to add around image), Magnification (factor by which to magnify each pixel before calculation of complexity), Method ("PerimeterLength" or "PolygonalLength", the method used to compute the perimeter), Threshold (a numerical threshold, or the method to use in thresholding operations after filtering), and Filter (None: no filter, "Gaussian" for Gaussian filter; "Sech" for a Hyperbolic Secant filter; or an array supplied directly as the filter kernel. For the gaussian and the Sech, the size of the filter is determined by the ViewingDistance option), ViewingDistance (in pixels, used to determine the radius of Gaussian and Sech filters), MorphologicalOperators (whether to use Mathematica's built in operators), Normalized (whether to normalize complexity by the value for a disk, 4\pi). Defaults are: \{\{Magnification\to Automatic, Pad\to Automatic, Verbose\to False, Threshold\to 0.5, Method\to \"PolygonalLength\", Filter\to \"Sech\", Normalized\to True, ViewingDistance\to 48/ArcTan[1°], MorphologicalOperators\to True, SechScale\to 2.16/60, GaussianScale\to 2.33/60, ReturnImages\to False\}\};

Options[PerimetricComplexity] =  
\{Magnification\to Automatic, Pad\to Automatic, Verbose\to False, Threshold\to 0.5, Method\to \"PolygonalLength\", Filter\to \"Sech\", Normalized\to True, ViewingDistance\to 48 / ArcTan[1°], MorphologicalOperators\to True, SechScale\to 2.16 / 60, GaussianScale\to 2.33 / 60, ReturnImages\to False\};

2/28/11 12:10:52 (Local) In[5]=
PerimetricComplexity[image_, opts___Rule] := Module[
{tmp, tmp0, tmp1, tmp2, tmp3, tmp4, tmp5, plength, area, 
mag, radius, pad, verbose, fdim, pc, norm, threshold, 
method, filter, vdist, pixelsperdegree, sechpixels, 
sechdegrees, morpho, sechscale, gscale, anglescale, 
rimages, images},
{mag, pad, verbose, threshold, method, filter, norm, vdist,
morphic, sechscale, gscale, rimages) = 
{Magnification, Pad, Verbose, Threshold, Method, Filter,
 Normalized, ViewingDistance, MorphologicalOperators,
 SechScale, GaussianScale, ReturnImages} /. {opts} /. 
Options[PerimetricComplexity];
anglescale = ArcTan[1. Degree];
If[mag == Automatic, mag = Switch[filter
 , None, 1
 , _?ListQ, 1
 , "Gaussian", Ceiling[2. / (vdist anglescale gscale)]
 , "Sech", Ceiling[3. / (vdist anglescale sechscale)]
 ]]

pixelsperdegree = mag vdist anglescale;
sechpixels = Ceiling[pixelsperdegree 3 sechscale];
If[pad == Automatic,
pad = If[filter == None, 2,
 Ceiling[pixelsperdegree sechscale]]];
If[EvenQ[sechpixels], sechpixels++]
sechdegrees = sechpixels / pixelsperdegree;
radius = 2 gscale pixelsperdegree / Sqrt[2 Pi];
tmp0 = Binarize[image];
tmp1 = If[mag > 1, ImageResize[tmp0, mag ImageDimensions[tmp0]],
 tmp0];
tmp2 = ImagePad[tmp1, pad];
tmp4 = Switch[filter
 , None, tmp2
 , _?ListQ, ImageConvolve[tmp2, filter]
 , "Gaussian", GaussianFilter[tmp2, radius]
 , "Sech",
 ImageConvolve[tmp2, SechKernel2D[sechpixels {1, 1},
 sechdegrees {1, 1}, sechscale]]
 ];

tmp5 = If[ImageType[tmp4] == "Bit", tmp4,
 If[NumericQ[threshold], Binarize[tmp4, threshold],
 Binarize[tmp4, Method -> threshold]]];

{length, area} = If[morphic,
 Total[
 Last /*
 (tst = ComponentMeasurements[MorphologicalComponents[tmp5],
 {method, "Count"}, CornerNeighbors -> False])]
 ,
 {PerimeterLength[tmp5, Switch[method, "PerimeterLength",
 PixelBorderLength, "PolygonalLength", PixelPathLength]],
 Total[ImageData[tmp5], 2]]};
(*Print["mag = ", mag, " ppd = ", pixelsperdegree,
 " sechpixels = ", sechpixels, " pad = ", pad];*)
tmp6 = MorphologicalPerimeter[tmp5, CornerNeighbors -> False];
images = {tmp2, tmp4, tmp5, ColorCombine[{tmp2, tmp6, tmp6}]}
If[verbose, (}
Print @ GraphicsRow[images, ImageSize -> 400];
}
{plength, area, (plength^2 / area) / If[norm, 4 Pi, 1]} //
If[rimages, Append[#, images], #] &

Here is an example.

PerimetricComplexity[charimage, Verbose -> True] // N

{1016.14, 5977., 13.7472}

- **PerimeterLength**

2/28/11 12:10:52 (Local) In[6]=

PerimeterLength::usage =
"PerimeterPathLength[image_, verbose_: False] Given a
binary image that represents a set of perimeters,
defined by connected white pixels, compute the
total length of the perimeters. It first locates
all foreground pixels. It then extracts the 3 x 3
neighborhood of each, and using PixelPathLength,it
computes half the distance to each of the pixels
two nearest foreground neighbors. The perimeter is
the sum of all these distances. If verbose is true,
it shows a tally of the neighborhoods and their
corresponding path lengths.";
PerimeterLength[image_, method_: PixelBorderLength, verbose_: False] :=
Module[{positions, neighborhoods, tally, perim},
  perim = Perimeter[image];
  positions = Position[ImageData[perim], 1];
  neighborhoods = ImageTake[Switch[method, PixelPathLength, perim, PixelBorderLength, image],
    Sequence @@ Transpose[{{# - 1, # + 1} & /@ positions};
  tally = Transpose[Tally[neighborhoods]];
  If[verbose, {
    Print["pixels = ", Length[positions]];
    Print @
    TableForm[Transpose[{tally[[1]], tally[[2]]},
      method /@ tally[[1]]]]];
  Total[tally[[2]] (method /@ tally[[1]])]
]

Here is an example.

test = ImagePad[Image[{{0, 1}, {1, 1}}, "Bit"], 1]
PerimeterLength[test, PixelBorderLength, True]

```
  pixels = 3
```

```
  1 3
  1 3
  1 2
```

- **PixelPathLength**

  PixelPathLength::usage = "PixelPathLength[image_] Given a 3x3 binary image, finds the coordinates of white pixels not at the center, and returns their mean distance from the center.";
Here is an example.

```math
PixelPathLength[image_] := Module[{tmp1, tmp2, tmp3},
   tmp1 = DeleteCases[Position[ImageData[image], 1], {2, 2}];
   tmp2 = (x - {2, 2}) & /@ tmp1;
   tmp3 = Sort[Norm /@ tmp2];
   Mean[Take[tmp3, 2]]
]
```

```
2/28/11 12:10:53 (Local) In[10]:=
PixelPathLength[tst]
```

```
1/2 (1 + \sqrt{2})
```

### PixelBorderLength

```
2/28/11 12:10:53 (Local) In[10]:=
PixelBorderLength::usage =
"PixelBorderLength[image_] Given a 3x3 binary image, counts the number of black neighbor (4-connected) pixels. This is a measure of the length of the exposed border of the foreground (white) pixel."
```

```
2/28/11 12:10:53 (Local) In[11]:=
PixelBorderLength[image_] :=
Total[1 - Flatten[ImageData[image]]][[{2, 4, 6, 8}]]
```

Here is an example.

```math
Framed[tst = Image[{{{0, 0, 0}, {0, 1, 1}, {1, 0, 0}}, "Bit"]
```

```
PixelBorderLength[tst]
```

```
2
```
- **SechKernel2D**

    SechKernel2D::usage = 
    "SechKernel2D[samples_,degrees_,scale_] Compute a convolution kernel defined by a Hyperbolic Secant (Sech) function of distance from the origin. samples is a list {height,width} of dimensions of the kernel in pixels, and degrees is a list of the corresponding dimensions in degrees of visual angle. scale defines the width of the kernel in degrees. The origin is defined as Floor[samples/2]. The kernel is normalized."

    2/28/11 12:26:19 (Local) In[38]:=
    SechKernel2D[samples_, degrees_, scale_] := 
    (# / Total[#2]) & @
    Sech[
    Pi Array[N[Norm[{##} degrees / samples]] & , samples,
    -Floor[samples / 2]] / scale]

Here is an example.

    ImageAdjust[Image[SechKernel2D[{32, 32}, {1, 1}, .4]]]

- **Perimeter**

    2/28/11 12:10:53 (Local) In[13]:=
    Perimeter[Image_] := Module[{padded, positions},
    padded = ImagePad[Image, 1];
    positions = Select[Position[ImageData[padded], 1],
    PixelBorderLength[ImageTake[padded, 
    Sequence @@ Transpose[{## - 1, ## + 1}]]] > 0 &];
    ImagePad[
    Image[ReplacePart[ImageData[padded] 0, # -> 1 & @@ positions],
    "Bit"], -1]]

- **Initializations**

    2/28/11 12:10:53 (Local) In[14]:=
    SetOptions[Graphics, ImageSize -> 128];

    2/28/11 12:10:53 (Local) In[15]:=
    SetOptions[ListLinePlot, Axes -> False, 
    BaseStyle -> {10, FontFamily -> "Helvetica", AbsolutePointSize[4]}, 
    Frame -> True, ImageSize -> 300];
The next cell, containing the definition of the pointspread function, is closed to suppress printing.

```
2/28/11 12:10:53 (Local) In[20]:=
jun = 整;
```

```
2/28/11 12:10:53 (Local) In[21]:=
yi = 以;
```

```
2/28/11 12:10:53 (Local) In[22]:=
chararray =
{
{{C, D, H, K, N, O, R, S, V, Z},
{个, 文, 么, 山, 少}
}};
```
Computing Complexity without using Mathematica’s Morphological Operators

Some readers may wish to write code in other languages to compute complexity. For that reason we provide an explanation here of how to compute complexity without using Mathematica’s morphological operators. This amounts to finding alternate methods for computing the perimeter. These are incorporated in the function `PerimeterLength`, defined above, and derived below. These functions can be exercised from within PerimetricComplexity by selecting the option `MorphologicalOperators->False`. This is useful mainly for testing.

Consider the following binary image.
The foreground area is easily obtained, since it is just the sum of all the white pixels:

\[
\text{Total[ImageData[jun], 2]} \quad 1467
\]

- **Using “PerimeterLength”**

As noted above, there are two definitions of the perimeter of a binary digital image. The first consists of the sum of the exposed pixel faces. To count the exposed faces we use the function \text{PixelBorderLength}. This takes a binary image of dimensions \{3,3\}, and counts the number of black 4-connected neighbors of the center pixel.

Here is an example.

\[
\text{Framed[test = Image[\{(0, 0, 0), (0, 1, 1), (1, 1, 1)\}, "Bit"]]} \\
\text{PixelBorderLength[test]} \quad 2
\]

In this example, the center pixel has only two black neighboring pixels.

The total perimeter can be obtained by applying this function to the \{3,3\} neighborhood of every white pixel in the image. Pixels in the interior will return a value of 0, so only perimeter pixels will contribute.

To implement this idea, we first identify the positions of all the white pixels.

\[
\text{positions = Position[ImageData[jun], 1];}
\]

Next we extract all the \{3,3\} neighborhoods.

\[
\text{neighborhoods = } \\
\text{ImageTake[jun, Sequence @@ Transpose[\{\# - 1, \# + 1\}] & /@ positions];}
\]

We can look at the first five.
Framed /@ neighborhoods[[Range[5]]]  
\[ \{\text{\includegraphics[width=0.1\textwidth]{image1.png}}, \text{\includegraphics[width=0.1\textwidth]{image2.png}}, \text{\includegraphics[width=0.1\textwidth]{image3.png}}, \text{\includegraphics[width=0.1\textwidth]{image4.png}}, \text{\includegraphics[width=0.1\textwidth]{image5.png}}\} \]

Framed /@ neighborhoods[[Range[5]]]  
\[ \{\text{\includegraphics[width=0.1\textwidth]{image1.png}}, \text{\includegraphics[width=0.1\textwidth]{image2.png}}, \text{\includegraphics[width=0.1\textwidth]{image3.png}}, \text{\includegraphics[width=0.1\textwidth]{image4.png}}, \text{\includegraphics[width=0.1\textwidth]{image5.png}}\} \]

We can also compute the pixel border length of the first five.

\begin{verbatim}
PixelBorderLength /@ neighborhoods[[Range[5]]]  
\{2, 1, 2, 2, 0\}
\end{verbatim}

In a large complex image, the same neighborhood might occur many times, so we perform a tally.

tally = Tally[neighborhoods];

We can take a look at the tally, along with the pixel border length for each type of neighborhood. Note that 850 cases consist of all white, drawn from the interior of the foreground, with 0 border length. Here we just look at the first 10 elements of the tally.
The total perimeter length will be the sum of all of the border lengths for all the collected neighborhoods. We use the tallied neighborhoods, so that the pixel border length of each type of neighborhood is computed only once.

\[
\text{Total}[(#[[2]] \text{PixelBorderLength}[#[[1]]] & @ tally)]
\]

606

Thus the complexity would be

\[
606^2 / 1467 / (4 \pi) \quad / / \text{N}
\]

19.9207
Using “PolygonalLength”

The second definition of the length of the perimeter is the sum of sides of a the polygon defined by the perimeter pixels considered as points in a lattice.

To extract the perimeter, we use a new function Perimeter, defined above. This duplicates a builtin Mathematica function. We verify that they yield the same results.

\[
\text{perimeter} = \text{Perimeter}[@\text{jun}]
\]

\[
\text{perimeter2} = \text{MorphologicalPerimeter}[@\text{jun}, \text{CornerNeighbors} \to \text{False}]
\]

\[
\text{ColorCombine}[@\text{jun}, \text{perimeter}, \text{perimeter2}]^
\]

We identify the positions of all of the pixels in the perimeter.

\[
\text{Length}[@\text{positions} = \text{Position}[@\text{ImageData}[@\text{perimeter}], 1]]
\]

498

We extract all of the 3x3 neighborhoods of pixels in the perimeter.

\[
\text{neighborhoods} = \text{ImageTake}[@\text{perimeter}, \text{Sequence} @@\text{Transpose}[\{# - 1, # + 1\}] & /@ \text{positions};
\]

Then we use a new function PixelPathLength. This looks at a 3x3 binary neighborhood, identi-
defines the two closest white pixels not at the center, and finds the distances from their centers to that of the central pixel. That is the path length corresponding to that neighborhood.

We apply this to all the perimeter pixels, and add up the results.

```
Total[PixelPathLength /@ neighborhoods] // N
```

547.084

We can verify this is the same perimeter length obtained from PerimetricComplexity using Mathematica’s morphological operators.

```
PerimetricComplexity[jun, Filter -> None,
Method -> "PolygonalLength"] // N
```

\{547.084, 1467.\,16.2356\}

**Acknowledgments**

I thank and blame Denis Pelli for introducing me to perimetric complexity [5]. I thank Dr. Cong Yu for providing the Chinese character optotypes [6]. I thank Albert Ahumada and Jeffrey Mulligan for useful discussions. I thank Larry Thibos for providing the wavefront data [16]. This work supported by NASA Space Human Factors Engineering WBS 466199.

**References**


About the Authors

Andrew B. Watson is the Senior Scientist for Vision Research at NASA. He is Editor-in-Chief of the Journal of Vision (http://journalofvision.org). He is the author of over 150 scientific papers and four patents. He is a Fellow of the Optical Society of America, the Association for Research in Vision and Ophthalmology, and the Society for Information Display.

Andrew B Watson
MS 262-2
NASA Ames Research Center
Moffett Field, CA 94035
hn://1/
andrew.b.watson@nasa.gov